Pensieve header: A free-Lie calculator.

```
Unprotect[NonCommutativeMultiply];
x_ ** 0 = 0; 0 ** y_ = 0;
(c_ * x_AW) ** y_ := Expand[c (x ** y)];
x_ ** (c_ * y_AW) := Expand[c (x ** y)];
x_Plus ** y_ := (# ** y) & /@ x;
x_ ** y_Plus := (x ** #) & /@ y;
AW[w1_String] ** AW[w2_String] := AW[w1 <> w2];
```

A Lyndon word is a word lexicographically smaller than all of its proper right factors; see http://katlas.math.toronto.edu/drorbn/bbs/show?shot=Chu-071214-182203.jpg

```
LyndonQ[AW[w_String]] := And @@ (
    OrderedQ[{w, #}] & /@ Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]
   );
AllWords[0, _List] = {AW[""]};
AllWords[n_ /; n > 0, ab_List] := AllWords[n, ab] = AW /@ Flatten[Outer[
      StringJoin[#1, #2] &,
      First /@ AllWords[n - 1, ab],
      ab
     ]];
AllLyndonWords[n_Integer, ab_List] := LW @@@ Select[AllWords[n, ab], LyndonQ];
AllLyndonWords[{n_}, ab_List] := Join @@ Table[AllLyndonWords[k, ab], {k, n}];
LyndonFactorization[LW[w_String] /; StringLength[w] == 1] := LW[w];
LyndonFactorization[LW[w_String] /; StringLength[w] > 1] := Module[
   {rf},
   rf = First[Sort[Table[StringDrop[w, i], {i, 1, StringLength[w] - 1}]]];
   LW /@ {StringDrop[w, -StringLength[rf]], rf}
  ];
LW[s_Symbol] := LW[ToString[s]];
LW[LW[w_]] := LW[w];
LW /: LW[x_] ≤ LW[y_] := OrderedQ[{x, y}];
LW /: x_LW ≥ y_LW := y ≤ x;
LW /: x_LW > y_LW := ! (x ≤ y);
LW /: x_LW < y_LW := ! (y ≤ x);
Format[LW[w_], StandardForm] := Defer[〈w〉];
〈w__〉 := LW[w];
LW[is__Integer] := LW[
   StringJoin @@ (StringTake["1234567890abcdefghijklmnopqrstuvwxyz", {#}] & /@ {is})];
Deg[LW[x_]] := StringLength[x];
```

```
{LyndonQ[AW@"abba"], LyndonQ[AW@"ababb"]}
```

```
{False, True}
```

```
{AllWords[3, {"1", "2"}], AllLyndonWords[{3}, {"1", "2"}]}
```

```
{{AW[111], AW[112], AW[121], AW[122], AW[211], AW[212], AW[221], AW[222]},
 {〈1〉, 〈2〉, 〈12〉, 〈112〉, 〈122〉}}
```

```
Table[Length[AllLyndonWords[k, {"1", "2"}]], {k, 10}]
```

```
{2, 1, 2, 3, 6, 9, 18, 30, 56, 99}
```

```
Table[Length[AllLyndonWords[k, {"1", "2", "3"}]], {k, 10}]
```

```
{3, 3, 8, 18, 48, 116, 312, 810, 2184, 5880}
```

```
TreeForm[LW["12122"] //. w_LW :> LyndonFactorization[w] /. LW[w_] :> w]
```




```
b[0, _] = 0; b[_, 0] = 0;
b[c_ * (x_AW | x_LW), y_] := Expand[c b[x, y]];
b[x_, c_ * (y_AW | y_LW)] := Expand[c b[x, y]];
b[x_Plus, y_] := b[#, y] & /@ x;
b[x_, y_Plus] := b[x, #] & /@ y;
b[w_AW, z_AW] := w ** z - z ** w;
b[w_LW, z_LW] := LWBracket[w, z];
```

```
LWBracket[w_LW, z_LW] := Which[
   (* If[Deg[w]+Deg[z]>4, Dialog[]]; *)
   w === z, 0,
   z < w, Expand[-b[z, w]],
   Deg[w] == 1, LW[First[w] <> First[z]],
   True, Module[{x, y},
    {x, y} = LyndonFactorization[w];
    If[y ≥ z,
     LW[First[w] <> First[z]],
     LWBracket[w, z] = b[x, LWBracket[y, z]] + b[LWBracket[x, z], y]
    ]
   ]
  ];
```

```
b[LW["112"], LW["122"]]
```

```
⟨112122⟩ + ⟨112212⟩
```

```
Outer[b, AllLyndonWords[{3}, {"1", "2"}],
  AllLyndonWords[{3}, {"1", "2"}]] // MatrixForm
```

0	⟨12⟩	⟨112⟩	⟨1112⟩	⟨1122⟩
−⟨12⟩	0	−⟨122⟩	−⟨1122⟩	−⟨1222⟩
−⟨112⟩	⟨122⟩	0	−⟨11212⟩	⟨12122⟩
−⟨1112⟩	⟨1122⟩	⟨11212⟩	0	⟨112122⟩ + ⟨112212⟩
−⟨1122⟩	⟨1222⟩	−⟨12122⟩	−⟨112122⟩ − ⟨112212⟩	0

```
Union[Flatten[Outer[(b[#1, #2] + b[#2, #1]) &,
   AllLyndonWords[{6}, {"1", "2"}], AllLyndonWords[{6}, {"1", "2"}]
  ]]]
```

{0}

```
Outer[(b[#1, b[#2, #3]] + b[#2, b[#3, #1]] + b[#3, b[#1, #2]]) &,
   AllLyndonWords[{5}, {"1", "2"}],
   AllLyndonWords[{5}, {"1", "2"}], AllLyndonWords[{5}, {"1", "2"}]
  ] // Flatten //
 Union
```

{0}

```
ad[x_][y_] := b[x, y];
MakeLieSeries[d_, l_List] := MakeLieSeries[d, #] & /@ l;
MakeLieSeries[d_, s_LieSeries] /; Length[s] ≤ d := s;
MakeLieSeries[d_, s_LieSeries] /; Length[s] > d := Take[s, d];
MakeLieSeries[d_, a_ → b_] := (a → MakeLieSeries[d, b]);
MakeLieSeries[d_, expr_] := LieSeries @@ Table[
    expr /. w_LW /; Deg[w] ≠ k → 0,
    {k, d}
   ];
MakeLieSeries[w_LW] := Append[LieSeries @@ Table[0, {Deg[w] - 1}], w];
LieSeries /: Expand[s_LieSeries] := Expand /@ s;
LieSeries /: Plus[ss__LieSeries] /; Length[{ss}] > 1 := Module[
   {l = Min[Length /@ {ss}]},
   LieSeries @@ Total[Take[List @@ #, l] & /@ {ss}]
  ];
LieSeries /: 0 * s_LieSeries := 0;
LieSeries /: c_?NumberQ * s_LieSeries := Expand[c * #] & /@ s;
LieSeries /: s_LieSeries + c_. * w_LW /; NumberQ[c] := Module[{d},
   d = Deg[w];
   If[Length[s] < d, 0,
    ReplacePart[s, d → s[[d]] + Expand[c * w]]
   ]
  ];
b[s1_LieSeries, s2_LieSeries] := Module[
   {d, k, m1, m2},
   m1 = 1 + LengthWhile[s1, # == 0 &];
   m2 = 1 + LengthWhile[s2, # == 0 &];
   LieSeries @@ Table[
     Sum[b[s1[[k]], s2[[d - k]]], {k, m1, d - m2}],
     {d, Min[m1 + Length[s2], m2 + Length[s1]]}
    ]
  ];
```

```
b[w_LW, s_LieSeries] := Join[
   LieSeries @@ Table[0, {Deg[w]}],
   ad[w] /@ s
  ];
b[s_LieSeries, w_LW] := Expand[-b[w, s]];
LieSeries /: EulerE[s_LieSeries] :=
  LieSeries @@ Expand[Range[Length[s]] * (List @@ s)];
OperatorSeries[f_, var_ → op_, d_][ψ_] := Module[
   {ser, as, ni, nf, t, l},
   ser = List @@ Series[f, {var, 0, d}];
   {as, ni, nf} = ser[[{3, 4, 5}]];
   t = Nest[op, ψ, ni];
   If[Head[t] === LieSeries,
    l = Length[t];
    Expand[as.NestList[MakeLieSeries[l, op[#]] &, t, nf - ni - 1]],
    Expand[as.NestList[op, t, nf - ni - 1]]
   ]
  ];
Ad[s_LieSeries] := OperatorSeries[E^(-ad), ad → ad[s], Length[s] - 1];
```

```
xw = ⟨x⟩; yw = ⟨y⟩;
OperatorSeries[E^(-ad), ad → ad[yw], 3][xw]
```

$$\langle x\rangle + \langle xy\rangle + \frac{\langle xyy\rangle}{2} + \frac{\langle xyyy\rangle}{6}$$

```
MakeLieSeries[5, OperatorSeries[E^(-ad), ad → ad[xw], 3][yw]]
```

$$\text{LieSeries}\left[\langle y\rangle,\ -\langle xy\rangle,\ \frac{\langle xxy\rangle}{2},\ -\frac{\langle xxxy\rangle}{6},\ 0\right]$$

```
BCH[1] = LieSeries[⟨"x"⟩ + ⟨"y"⟩];
BCH[n_] := BCH[n] = Module[
   {bch, t1, t2},
   bch = Append[BCH[n - 1], 0];
   t1 =
    MakeLieSeries[n, ⟨"y"⟩ + OperatorSeries[E^(-ad), ad → ad[⟨"y"⟩], n - 1][⟨"x"⟩]];
   t2 = OperatorSeries[(1 - E^(-ad)) / ad, ad → ad[bch], n - 1][EulerE[bch]];
   bch + (t1 - t2) / n
  ]
```

```
BCH[2]
```

$$\text{LieSeries}\left[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2}\right]$$

```
BCH[8]
```

$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle,\ \frac{\langle xy\rangle}{2},\ \frac{\langle xxy\rangle}{12} + \frac{\langle xyy\rangle}{12},\ \frac{\langle xxyy\rangle}{24},$$
$$-\frac{\langle xxxxy\rangle}{720} + \frac{\langle xxxyy\rangle}{180} + \frac{\langle xxyxy\rangle}{360} + \frac{\langle xxyyy\rangle}{180} + \frac{\langle xyxyy\rangle}{120} - \frac{\langle xyyyy\rangle}{720},$$
$$-\frac{\langle xxxxyy\rangle}{1440} + \frac{\langle xxxyxy\rangle}{720} + \frac{\langle xxxyyy\rangle}{360} + \frac{\langle xxyxyy\rangle}{240} - \frac{\langle xxyyyy\rangle}{1440},$$
$$\frac{\langle xxxxxxy\rangle}{30\,240} - \frac{\langle xxxxxyy\rangle}{5040} + \frac{\langle xxxxyxy\rangle}{10\,080} + \frac{\langle xxxxyyy\rangle}{3780} + \frac{\langle xxxyxxy\rangle}{10\,080} + \frac{\langle xxxyxyy\rangle}{1680} +$$
$$\frac{\langle xxxyyxy\rangle}{1260} + \frac{\langle xxxyyyy\rangle}{3780} + \frac{\langle xxyxxyy\rangle}{2016} - \frac{\langle xxyxyxy\rangle}{5040} + \frac{13\,\langle xxyxyyy\rangle}{15\,120} + \frac{\langle xxyyxyy\rangle}{10\,080} -$$
$$\frac{\langle xxyyyxy\rangle}{1512} - \frac{\langle xxyyyyy\rangle}{5040} + \frac{\langle xyxyxyy\rangle}{1260} - \frac{\langle xyxyyyy\rangle}{2016} - \frac{\langle xyyxyyy\rangle}{5040} + \frac{\langle xyyyyyy\rangle}{30\,240},$$
$$\frac{\langle xxxxxxyy\rangle}{60\,480} - \frac{\langle xxxxxyxy\rangle}{15\,120} - \frac{\langle xxxxxyyy\rangle}{10\,080} + \frac{\langle xxxxyxxy\rangle}{20\,160} - \frac{\langle xxxxyxyy\rangle}{20\,160} + \frac{\langle xxxxyyxy\rangle}{2520} +$$
$$\frac{23\,\langle xxxxyyyy\rangle}{120\,960} + \frac{\langle xxxyxxyy\rangle}{4032} - \frac{\langle xxxyxyxy\rangle}{10\,080} + \frac{13\,\langle xxxyxyyy\rangle}{30\,240} + \frac{\langle xxxyyxyy\rangle}{20\,160} -$$
$$\frac{\langle xxxyyyxy\rangle}{3024} - \frac{\langle xxxyyyyy\rangle}{10\,080} + \frac{\langle xxyxyxyy\rangle}{2520} - \frac{\langle xxyxyyyy\rangle}{4032} - \frac{\langle xxyyxyyy\rangle}{10\,080} + \frac{\langle xxyyyyyy\rangle}{60\,480}\Big]$$

```
LieMorphism[rules_List] :=
  LieMorphism[rules] = LieMorphism[Unique[LieMorphism], rules];
LieMorphism[mor_Symbol, rules_List] := (
   mor[] = rules;
   (mor[w_LW] /; Deg[w] == 1) := (mor[w] = w /. rules);
   mor[w_LW] := (mor[w] = b @@ (mor /@ LyndonFactorization[w]));
   mor[s_LieSeries] := Module[
     {l = Length[s]},
     MakeLieSeries[l, Sum[
       LieMorphism[MakeLieSeries[l - k + 1, rules]][s[[k]]],
       {k, l}
      ]]
    ];
   mor[expr_] := Expand[expr /. {s_LieSeries ⧴ mor[s], w_LW ⧴ mor[w]}];
   mor
  );
BCH[n_, x_, y_] := LieMorphism[{LW["x"] → x, LW["y"] → y}][BCH[n]];
BCH[s1_LieSeries, s2_LieSeries] := BCH[Min[Length /@ {s1, s2}], s1, s2];
```

```
{n = 4,
 t1 = BCH[n, LW["x"], BCH[n, LW["y"], LW["z"]]],
 t2 = BCH[n, BCH[n, LW["x"], LW["y"]], LW["z"]],
 t1 == t2}
```

$$\Big\{4, \text{LieSeries}\Big[\langle x\rangle + \langle y\rangle + \langle z\rangle, \frac{\langle xy\rangle}{2} + \frac{\langle xz\rangle}{2} + \frac{\langle yz\rangle}{2}, \frac{\langle xxy\rangle}{12} + \frac{\langle xxz\rangle}{12} + \frac{\langle xyy\rangle}{12} + \frac{\langle xyz\rangle}{3} + \frac{\langle xzy\rangle}{6} + \frac{\langle xzz\rangle}{12} + \frac{\langle yyz\rangle}{12} + \frac{\langle yzz\rangle}{12}, \frac{\langle xxyy\rangle}{24} + \frac{\langle xxyz\rangle}{12} + \frac{\langle xxzy\rangle}{12} + \frac{\langle xxzz\rangle}{24} + \frac{\langle xyyz\rangle}{12} + \frac{\langle xyzy\rangle}{12} + \frac{\langle xyzz\rangle}{12} + \frac{\langle xzyz\rangle}{12} + \frac{\langle yyzz\rangle}{24}\Big],$$
$$\text{LieSeries}\Big[\langle x\rangle + \langle y\rangle + \langle z\rangle, \frac{\langle xy\rangle}{2} + \frac{\langle xz\rangle}{2} + \frac{\langle yz\rangle}{2}, \frac{\langle xxy\rangle}{12} + \frac{\langle xxz\rangle}{12} + \frac{\langle xyy\rangle}{12} + \frac{\langle xyz\rangle}{3} + \frac{\langle xzy\rangle}{6} + \frac{\langle xzz\rangle}{12} + \frac{\langle yyz\rangle}{12} + \frac{\langle yzz\rangle}{12}, \frac{\langle xxyy\rangle}{24} + \frac{\langle xxyz\rangle}{12} + \frac{\langle xxzy\rangle}{12} + \frac{\langle xxzz\rangle}{24} + \frac{\langle xyyz\rangle}{12} + \frac{\langle xyzy\rangle}{12} + \frac{\langle xyzz\rangle}{12} + \frac{\langle xzyz\rangle}{12} + \frac{\langle yyzz\rangle}{24}\Big], \text{True}\Big\}$$

```
Timing[{n = 10,
  Length /@ (t1 = BCH[n, LW["x"], BCH[n, LW["y"], LW["z"]]]),
  Length /@ (t2 = BCH[n, BCH[n, LW["x"], LW["y"]], LW["z"]]),
  t1 == t2}]
```

```
{8.19, {10, LieSeries[3, 3, 8, 9, 48, 82, 312, 622, 2069, 4597],
  LieSeries[3, 3, 8, 9, 48, 82, 312, 622, 2069, 4597], True}}
```

```
Timing[{n = 10,
  Length /@ (t1 = BCH[n, LW["x"], BCH[n, LW["y"], LW["z"]]]),
  Length /@ (t2 = BCH[n, BCH[n, LW["x"], LW["y"]], LW["z"]]),
  t1 == t2}]
```

```
{0.624, {10, LieSeries[3, 3, 8, 9, 48, 82, 312, 622, 2069, 4597],
  LieSeries[3, 3, 8, 9, 48, 82, 312, 622, 2069, 4597], True}}
```

```
ι[w_LW] /; Deg[w] == 1 := AW @@ w;
ι[w_LW] := ι[w] = b @@ (ι /@ LyndonFactorization[w]);
ι[s_LieSeries] := Prepend[ι /@ (ASeries @@ s), 0];
ι[expr_] := Expand[expr /. w_LW :→ ι[w]];
```

```
t1 = ι[BCH[3]]
```

$$\text{ASeries}\Big[0, \text{AW}[x] + \text{AW}[y], \frac{\text{AW}[xy]}{2} - \frac{\text{AW}[yx]}{2}, \frac{\text{AW}[xxy]}{12} - \frac{\text{AW}[xyx]}{6} + \frac{\text{AW}[xyy]}{12} + \frac{\text{AW}[yxx]}{12} - \frac{\text{AW}[yxy]}{6} + \frac{\text{AW}[yyx]}{12}\Big]$$

```
ASeries /: Expand[s_ASeries] := Expand /@ s;
ASeries /: Plus[ss__ASeries] := Module[
   {l = Min[Length /@ {ss}]},
   ASeries @@ Total[Take[List @@ #, l] & /@ {ss}]
  ];
ASeries /: c_ * s_ASeries := Expand[c * #] & /@ s;
s1_ASeries ** s2_ASeries := Module[
   {d, k, m1, m2},
   m1 = LengthWhile[s1, # == 0 &];
   m2 = LengthWhile[s2, # == 0 &];
   ASeries @@ Table[
     Sum[s1[[k + 1]] ** s2[[d - k + 1]], {k, m1, d - m2}],
     {d, 0, Min[m1 + Length[s2] - 1, m2 + Length[s1] - 1]}
    ]
  ];
ASeries /: EulerE[s_ASeries] :=
  ASeries @@ Expand[Range[{0, 1 + Length[s]}] * (List @@ s)];
```

```
ASeries[AW[""], 0, 0, 0] + t1 + t1 ** t1 / 2 + t1 ** t1 ** t1 / 6
```

ASeries$\left[\text{AW}[], \text{AW}[x] + \text{AW}[y], \frac{\text{AW}[xx]}{2} + \text{AW}[xy] + \frac{\text{AW}[yy]}{2}, \frac{\text{AW}[xxx]}{6} + \frac{\text{AW}[xxy]}{2} + \frac{\text{AW}[xyy]}{2} + \frac{\text{AW}[yyy]}{6}\right]$

```
ASeries[AW[""], AW["x"], AW["xx"] / 2, AW["xxx"] / 6] **
 ASeries[AW[""], AW["y"], AW["yy"] / 2, AW["yyy"] / 6]
```

ASeries$\left[\text{AW}[], \text{AW}[x] + \text{AW}[y], \frac{\text{AW}[xx]}{2} + \text{AW}[xy] + \frac{\text{AW}[yy]}{2}, \frac{\text{AW}[xxx]}{6} + \frac{\text{AW}[xxy]}{2} + \frac{\text{AW}[xyy]}{2} + \frac{\text{AW}[yyy]}{6}\right]$

```
σ[y_LW, w_LW] /; Deg[y] == 1 := σ[y, w] = Which[
    y === w, AW[""],
    Deg[w] === 1, 0,
    True, Module[{w1, w2},
     {w1, w2} = LyndonFactorization[w];
     ι[w1] ** σ[y, w2] - ι[w2] ** σ[y, w1]
    ]
   ];
σ[y_, expr_] := Expand[expr /. w_LW :> σ[LW[y], w]] /. LieSeries → ASeries;
```

```
(# -> σ[1, #]) & /@ AllLyndonWords[{5}, {"1", "2"}]
```

{⟨1⟩ → AW[], ⟨2⟩ → 0, ⟨12⟩ → -AW[2], ⟨112⟩ → -2 AW[12] + AW[21], ⟨122⟩ → AW[22],
⟨1112⟩ → -3 AW[112] + 3 AW[121] - AW[211], ⟨1122⟩ → 2 AW[212] - AW[221],
⟨1222⟩ → -AW[222], ⟨11112⟩ → -4 AW[1112] + 6 AW[1121] - 4 AW[1211] + AW[2111],
⟨11122⟩ → -AW[1122] + 4 AW[1212] - AW[1221] - 2 AW[2121] + AW[2211],
⟨11212⟩ → -AW[1122] + 4 AW[1212] - AW[1221] - 3 AW[2112] + AW[2121],
⟨11222⟩ → -2 AW[1222] + 3 AW[2122] - 3 AW[2212] + AW[2221],
⟨12122⟩ → 2 AW[1222] - 3 AW[2122] + AW[2212], ⟨12222⟩ → AW[2222]}

```
xw = ⟨"x"⟩; yw = ⟨"y"⟩;
{σ[xw, BCH[5, xw, yw]], σ[yw, BCH[5, xw, yw]]} /. AW[s_] ⧴ s
```

$$\Big\{\text{ASeries}\Big[, -\frac{y}{2}, -\frac{xy}{6}+\frac{yx}{12}+\frac{yy}{12}, \frac{yxy}{12}-\frac{yyx}{24}, \frac{xxxy}{180}-\frac{xxyx}{120}-\frac{xxyy}{120}+\frac{xyxx}{180}+\frac{xyxy}{30}-\frac{xyyx}{120}+\frac{xyyy}{180}-\frac{yxxx}{720}-\frac{yxxy}{120}-\frac{yxyx}{120}-\frac{yxyy}{120}+\frac{yyxx}{180}-\frac{yyxy}{120}+\frac{yyyx}{180}-\frac{yyyy}{720}\Big],$$

$$\text{ASeries}\Big[, \frac{x}{2}, \frac{xx}{12}+\frac{xy}{12}-\frac{yx}{6}, \frac{xxy}{24}-\frac{xyx}{12}, -\frac{xxxx}{720}+\frac{xxxy}{180}-\frac{xxyx}{120}+\frac{xxyy}{180}-\frac{xyxx}{120}-\frac{xyxy}{120}-\frac{xyyx}{120}-\frac{xyyy}{720}+\frac{yxxx}{180}-\frac{yxxy}{120}+\frac{yxyx}{30}+\frac{yxyy}{180}-\frac{yyxx}{120}-\frac{yyxy}{120}+\frac{yyyx}{180}\Big]\Big\}$$

```
σ[xw, BCH[8, xw, yw]] /. AW[s_] ⧴ s
```

$$\text{ASeries}\Big[, -\frac{y}{2}, -\frac{xy}{6}+\frac{yx}{12}+\frac{yy}{12}, \frac{yxy}{12}-\frac{yyx}{24},$$
$$\frac{xxxy}{180}-\frac{xxyx}{120}-\frac{xxyy}{120}+\frac{xyxx}{180}+\frac{xyxy}{30}-\frac{xyyx}{120}+\frac{xyyy}{180}-\frac{yxxx}{720}-\frac{yxxy}{120}-\frac{yxyx}{120}-\frac{yxyy}{120}+\frac{yyxx}{180}-\frac{yyxy}{120}+\frac{yyyx}{180}-\frac{yyyy}{720},$$
$$-\frac{yxxxy}{360}+\frac{yxxyx}{240}+\frac{yxxyy}{240}-\frac{yxyxx}{360}-\frac{yxyxy}{60}+\frac{yxyyx}{240}-\frac{yxyyy}{360}+\frac{yyxxx}{1440}+\frac{yyxxy}{240}+\frac{yyxyx}{240}+\frac{yyxyy}{240}-\frac{yyyxx}{360}-\frac{yyyxy}{360}+\frac{yyyyx}{1440},$$
$$-\frac{xxxxxy}{5040}+\frac{xxxxyx}{2016}+\frac{xxxxyy}{2016}-\frac{xxxyxx}{1512}-\frac{xxxyxy}{630}-\frac{xxxyyx}{5040}-\frac{xxxyyy}{1512}+\frac{xxyxxx}{2016}+\frac{xxyxxy}{840}+\frac{xxyxyx}{840}+\frac{xxyxyy}{840}-\frac{xxyyxx}{5040}+\frac{xxyyxy}{840}-\frac{xxyyyx}{5040}+\frac{xxyyyy}{2016}-\frac{xyxxxx}{5040}-\frac{xyxxxy}{630}+\frac{xyxxyx}{840}+\frac{xyxxyy}{840}-\frac{xyxyxx}{630}-\frac{xyxyxy}{140}+\frac{xyxyyx}{840}-\frac{xyxyyy}{630}+\frac{xyyxxx}{2016}+\frac{xyyxxy}{840}+\frac{xyyxyx}{840}+\frac{xyyxyy}{840}-\frac{xyyyxx}{1512}-\frac{xyyyxy}{630}+\frac{xyyyyx}{2016}-\frac{xyyyyy}{5040}+\frac{yxxxxx}{30240}+\frac{yxxxxy}{2016}-\frac{yxxxyx}{5040}-\frac{yxxxyy}{5040}-\frac{yxxyxx}{5040}+\frac{yxxyxy}{840}-\frac{yxxyyx}{1120}-\frac{yxxyyy}{5040}+\frac{yxyxxx}{2016}+\frac{yxyxxy}{840}+\frac{yxyxyx}{840}+\frac{yxyxyy}{840}-\frac{yxyyxx}{5040}+\frac{yxyyxy}{840}-\frac{yxyyyx}{5040}+\frac{yxyyyy}{2016}-\frac{yyxxxx}{5040}-\frac{yyxxxy}{5040}-\frac{yyxxyx}{1120}-\frac{yyxxyy}{1120}-\frac{yyxyxx}{5040}+\frac{yyxyxy}{840}-\frac{yyxyyx}{1120}-\frac{yyxyyy}{5040}+\frac{yyyxxx}{3780}-\frac{yyyxxy}{5040}-\frac{yyyxyx}{5040}-\frac{yyyxyy}{5040}+\frac{yyyyxx}{3780}+\frac{yyyyxy}{2016}-\frac{yyyyyx}{5040}+\frac{yyyyyy}{30240},$$
$$\frac{yxxxxxy}{10080}-\frac{yxxxxyx}{4032}-\frac{yxxxxyy}{4032}+\frac{yxxxyxx}{3024}+\frac{yxxxyxy}{1260}+\frac{yxxxyyx}{10080}+\frac{yxxxyyy}{3024}-\frac{yxxyxxx}{4032}-\frac{yxxyxxy}{1680}-\frac{yxxyxyx}{1680}-\frac{yxxyxyy}{1680}+\frac{yxxyyxx}{10080}-\frac{yxxyyxy}{1680}+\frac{yxxyyyx}{10080}-\frac{yxxyyyy}{4032}+\frac{yxyxxxx}{10080}+\frac{yxyxxxy}{1260}-\frac{yxyxxyx}{1680}-\frac{yxyxxyy}{1680}+\frac{yxyxyxx}{1260}+\frac{yxyxyxy}{280}-\frac{yxyxyyx}{1680}+\frac{yxyxyyy}{1260}-\frac{yxyyxxx}{4032}-\frac{yxyyxxy}{1680}-\frac{yxyyxyx}{1680}-\frac{yxyyxyy}{1680}+\frac{yxyyyxx}{3024}+\frac{yxyyyxy}{1260}-\frac{yxyyyyx}{4032}+\frac{yxyyyyy}{10080}-\frac{yyxxxxx}{60480}-\frac{yyxxxxy}{4032}+\frac{yyxxxyx}{10080}+\frac{yyxxxyy}{10080}+\frac{yyxxyxx}{10080}-\frac{yyxxyxy}{1680}+\frac{yyxxyyx}{2240}+\frac{yyxxyyy}{10080}-\frac{yyxyxxx}{4032}-\frac{yyxyxxy}{1680}-\frac{yyxyxyx}{1680}-\frac{yyxyxyy}{1680}+\frac{yyxyyxx}{10080}-\frac{yyxyyxy}{1680}+\frac{yyxyyyx}{10080}-\frac{yyxyyyy}{4032}+\frac{yyyxxxx}{10080}+\frac{yyyxxxy}{3024}+\frac{yyyxxyx}{10080}+\frac{yyyxxyy}{10080}+\frac{yyyxyxx}{3024}+\frac{yyyxyxy}{1260}+\frac{yyyxyyx}{10080}+\frac{yyyxyyy}{3024}-\frac{23\,yyyyxxx}{120960}-\frac{yyyyxxy}{4032}-\frac{yyyyxyx}{4032}-\frac{yyyyxyy}{4032}+\frac{yyyyyxx}{10080}+\frac{yyyyyxy}{10080}-\frac{yyyyyyx}{60480}\Big]$$